//OreStages.startStone();

OreStages.startSend();

OreStages.startGravel();

//获取钻石
<recipetype:create:mixing>.addRecipe("meng_basics_diamond",
	"superheated",
	<item:minecraft:diamond>*1,
	[<item:minecraft:coal_block> * 16],
	[<fluid:minecraft:lava>,<fluid:minecraft:water>]
);
<recipetype:create:sequenced_assembly>.addRecipe(<recipetype:create:sequenced_assembly>.builder("meng_create_diamond")
	.transitionTo(<item:minecraft:coal>.definition)
	.require(<item:minecraft:coal_block>)
	.loops(3)
	.addOutput(<item:minecraft:diamond>, 0.5)
	.addOutput(<item:morecharcoal:coal_chunk>, 1)
	.addOutput(<item:minecraft:air>,2)
	.addStep(<recipetype:create:pressing>.factory(),(rb) => rb.duration(200))
	.addStep(<recipetype:create:cutting>.factory(), (rb) => rb.duration(100))
	.addStep(<recipetype:create:deploying>.factory(), (rb) => rb.require(<item:minecraft:coal_block>))
);

/*煤炭的获取*/
<recipetype:create:sandpaper_polishing>.addRecipe("meng_basics_coal",<item:minecraft:coal>,<item:minecraft:charcoal>);

/*各种种子的获取*/
<recipetype:create:milling>.addRecipe("meng_basics_seed",[<item:minecraft:wheat_seeds> % 60,<item:minecraft:beetroot_seeds> % 40,
	<item:immersiveengineering:seed> % 20,<item:minecraft:pumpkin_seeds> % 10,<item:minecraft:melon_seeds> % 10,
	<item:minecraft:brown_mushroom> % 5,<item:minecraft:red_mushroom> % 5
],<item:minecraft:grass>);
<recipetype:create:milling>.addRecipe("meng_basics_seed2",[<item:minecraft:wheat_seeds> % 60,<item:minecraft:beetroot_seeds> % 40,
	<item:immersiveengineering:seed> % 20,<item:minecraft:pumpkin_seeds> % 10,<item:minecraft:melon_seeds> % 10,
	<item:minecraft:brown_mushroom> % 5,<item:minecraft:red_mushroom> % 5
],<item:minecraft:tall_grass>);

/*铅、镍的获取*/
<recipetype:create:crushing>.removeRecipe(<item:create:crushed_copper_ore>);
<recipetype:create:crushing>.addRecipe("copper_get_ore_meng",[
	<item:create:crushed_copper_ore>,<item:immersiveengineering:nugget_lead> % 20,<item:immersiveengineering:nugget_nickel> % 20,
	<item:create:crushed_copper_ore> * 2 % 30 , <item:minecraft:cobblestone> % 12
],<item:create:copper_ore>,250);

/*铝、银的获取*/
<recipetype:create:crushing>.removeRecipe(<item:create:crushed_iron_ore>);
<recipetype:create:crushing>.addRecipe("iron_get_ore_meng",[
	<item:create:crushed_iron_ore>,<item:immersiveengineering:nugget_aluminum> % 20,<item:immersiveengineering:nugget_silver> % 20,
	<item:create:crushed_iron_ore> * 2 % 30 , <item:minecraft:cobblestone> % 12
],<item:minecraft:iron_ore>,250);